Physiologically based pharmacokinetic modeling of human exposure to perfluorooctanoic acid suggests historical non drinking-water exposures are important for predicting current serum

concentrations - SUPPLEMENTARY INFORMATION

Rachel Rogers Worley<sup>1,2</sup>, Xiaoxia Yang<sup>3</sup>, and Jeffrey Fisher<sup>2,3</sup>

 Division of Community Health Investigations, Agency for Toxic Substance and Disease Registry, Atlanta, GA, USA

2) Interdisciplinary Toxicology Program, University of Georgia, Athens, GA, USA

3) National Center for Toxicological Research, Food and Drug Administration, Jefferson, AR, USA

Corresponding Author:

Rachel Rogers Worley

ATSDR Division of Community Health Investigations

4770 Buford Highway, MS F59

Atlanta, GA 30341

770-488-1549

RWorley@cdc.gov

## **PBPK Model for Human Exposure to PFOA**

Model code was written and simulations were performed using AcslX modeling software (AEgis Technologies, Huntsville, AL, version 3.1.4.2).

```
PROGRAM PBPK Model for PFOA in Adult Humans.
! Created by R. Worley scaled from PFOA Adult Rat Model
! Model includes compartments for GI Tract, Liver, Rest of Body (Tis), Kidney Blood, Proximal Tubule Cells,
Filtrate, and Plasma.
! Exposure route is oral ingestion (into gut compartment)
! Excretion pathways are feces and urine.
! Units of concentration in ug/L.
  INITIAL
CONSTANT tstart = 0.0
  ! Physiological Parameters
    CONSTANT BW = 82.3 !bodyweight (kg); EPA Factors Handbook, 2011
    ! Cardiac Output and Bloodflow (as fraction of cardiac output)
    CONSTANT QCC = 12.5 !cardiac output in L/h/kg^0.75; Brown 1997, Forsyth 1968
    CONSTANT QLC = 0.25 !fraction blood flow to liver; Brown 1997, Fisher 2000
    CONSTANT QKC = 0.175 !fraction blood flow to kidney; Brown 1997, Forsyth 1968
    CONSTANT Htc = 0.44 !hematocrit for the human; Davies and Morris 1993, Brown 1997
    ! Tissue Volumes
    CONSTANT VplasC = 0.0428 !fraction vol. of plasma (L/kg BW); Davies 1993
    CONSTANT VLC = 0.026 !fraction vol. of liver (L/kg BW); Brown 1997
    CONSTANT VKC = 0.004 !fraction vol. of kidney (L/kg BW); Brown 1997
    CONSTANT VfilC = 0.0004 !fraction vol. of filtrate (L/kg BW)
    CONSTANT VPTCC = 1.35e-4 !vol. of proximal tubule cells (L/g kidney) (60 million PTC cells/gram kidney, 1
PTC = 2250 \text{ um}3) \text{ CHECK}
  ! Chemical Specific Parameters
    CONSTANT MW = 414.07 !PFOA molecular mass (q/mol)
```

CONSTANT Free = 0.02 !free fraction in plasma; Loccisano, 2011

!Kidney Transport Parameters

CONSTANT Vmax\_baso\_invitro = 439.2 !Vmax of basolateral transporter (pmol/mg protein/min); averaged in vitro value of OAT1 and OAT3 from Nakagawa, 2007

CONSTANT  $Km_baso = 20100.0 \; !Km \; of basolateral transporter (ug/L; average of OAT1 and OAT3 from Nakagawa et. al, 2007$ 

CONSTANT Vmax\_apical\_invitro = 37400.0 !Vmax of apical transporter (pmol/mg protein/min); invitro value for OAT4 from Yang, 2010.

CONSTANT  $Km_{apical} = 77500.0$  !Km of apical transporter (ug/L); averaged in vitro value for OAT4 and URAT1 from Yang, 2010.

CONSTANT RAFbaso = 1.0 !relative activity factor, basolateral transporters; fit to data CONSTANT RAFapi = 0.0007 !relative activity factor, apical transporters; fit to data

CONSTANT protein = 2.0e-6 !amount of protein in proximal tubule cells (mg protein/proximal tubule cell); Addis, 1936

CONSTANT GFRC = 24.19 !glomerular filtration rate (L/hr/kg kidney); Corley, 2005

## !Partition Coefficients

CONSTANT PL = 1.03 !liver:blood (from human cadaver data, Fabrega, 2014)

CONSTANT PK = 1.17 !kidney:blood (from human cadaver data, Fabrega, 2014)

CONSTANT PR = 0.11 !rest of body:blood (from rat tissue data, Kudo, 2007)

## !rate constants

CONSTANT kdif = 0.001 !diffusion rate from proximal tubule cells (L/h)

CONSTANT kabsc = 2.12 !rate of absorption of chemical from small intestine to liver  $(1/(h*BW^-0.25))$  (fit to data)

CONSTANT kunabsc = 7.06e-5 !rate of unabsorbed dose to appear in feces  $(1/(h*BW^-0.25))$  (fit to data)

CONSTANT GEC = 3.5 !gastric emptying time  $(1/(h*BW^-0.25))$ ; from Yang, 2014

CONSTANT k0C = 1.0 !rate of uptake from the stomach into the liver  $(1/(h*BW^-0.25))$ 

CONSTANT keffluxc = 0.1 !rate of clearance of PFOA from proximal tubule cells into blood ( $1/(h*BW^-0.25)$ )); fit to data

CONSTANT kbilec = 0.0001 !biliary elimination rate ((male); liver to feces storage ( $1/(h*BW^-0.25)$ ); fit to data

```
CONSTANT kurinec = 0.063 !rate of urine elimination from urine storage (male) (1/(h*BW^{-0.25})) (fit to
data)
    CONSTANT kvoid = 0.06974 !daily urine volume rate (L/hr); Van Haarst, 2004
    ! Exposure Parameters
    countdw = 0.0
   hourofday = 0.0
    dayofweek = 0.0
  END ! INITIAL
  DYNAMIC
    ALGORITHM IALG = 2
              NSTP = 10
    NSTEPS
    MAXTERVAL MAXT = 1.0e9
    MINTERVAL MINT = 1.0e-9
    CINTERVAL CINT = 100.0
     DERIVATIVE
   !Scaled Parameters
    !Cardiac output and blood flows
    QC = QCC*(BW**0.75)*(1-Htc) !cardiac output in L/h; adjusted for plasma
    QK = (QKC*QC) !plasma flow to kidney (L/h)
    QL = (QLC*QC) !plasma flow to liver (L/h)
    QR = QC - QK - QL !plasma flow to rest of body (L/h)
    QBal = QC - (QK + QL + QR) !Balance check of blood flows; should equal zero
    !Tissue Volumes
                      !volume of plasma (L)
    VPlas = VplasC*BW
    VK = VKC*BW !volume of kidney (L)
   MK = VK*1.0*1000 !mass of kidney (q); based on density of kidney = 1.0 q/mL
   VKb = VK*0.16 !volume of blood in the kidney (L); fraction blood volume of kidney (0.16) from Brown,
1997
    Vfil = VfilC*BW !volume of filtrate (L)
```

```
VL = VLC*BW !volume of liver (L)
   ML = VL*1.05*1000 !mass of liver (q); based on density of liver = 1.05 q/mL, Overmeyer 1987
   VR = (0.93*BW) - VPlas - VPTC - Vfil - VL !volume of remaining tissue (L) [Note: VKb not included as
already accounted for in VPlas];
   VBal = (0.93*BW) - (VR + VL + VPTC + VFil + VPlas) !Balance check of tissue volumes; should equal zero
    !Kidney Parameters
   PTC =VKC*1000*6e7 !number of PTC (cells/kg BW) (based on 60 million PTC/gram kidney); assuming density
of 1 kg/L
   VPTC = VK*1000*VPTCC !volume of proximal tubule cells (L)
   MPTC = VPTC*1000 !mass of the proximal tubule cells (g) (assuming density 1 kg/L)
   Vmax basoC = (Vmax baso invitro*RAFbaso*PTC*protein*60*(MW/1e12)*1000000)!Vmax of basolateral
transporters (ug/h/kg BW ^0.75)
   Vmax apicalC = (Vmax apical invitro*RAFapi*PTC*protein*60*(MW/1e12)*1000000) !Vmax of basolateral
transporters (ug/h/kg BW ^0.75)
   Vmax baso = Vmax basoC*BW**0.75 ! (ug/h)
   VMax apical = Vmax apicalC*BW**0.75 !(ug/h)
   kbile = kbilec*BW**(-0.25) !biliary elimination; liver to feces storage (/h)
   kurine = kurinec*BW**(-0.25) ! urinary elimination, from filtrate (/h)
   kefflux = keffluxc*BW**(-0.25) !efflux clearance rate, from PTC to blood (/h)
   GFR = GFRC*VK !glomerular filtration rate, scaled to mass of kidney(in kg)(L/h)
    !GI Tract Parameters
   kabs = kabsc*BW**(-0.25) !rate of absorption of chemical from small intestine to liver (/h)
   kunabs = kunabsc*BW**(-0.25) !rate of unabsorbed dose to appear in feces (/h)
   GE = GEC/BW^{**}(-0.25) !gastric emptying time (/h)
   k0 = KOC/BW^*(-0.25) !rate of uptake from the stomach into the liver (/h)
! Exposure Parameters
  day = t/24
 year = day/365
  !Drinking Water
  CONSTANT backgrounddw = 0.04 ! Drinking water concentration (ug/L or ppb)
 CONSTANT exposeddw = 0.04 !Contaminated drinking water concentration (ug/L or ppb)
```

```
CONSTANT dwtotal = 1.36 ! daily drinking water consumption (L), from the EPA Factors Handbook (2011)
  CONSTANT drinks = 4.0 !Total drinks per day (drink)
  CONSTANT tlendw = 0.25 !Length of time spent drinking each drinking event (hrs/drink)
  CONSTANT tbackground = 438000.0 !duration of exposure to background dw concentration (hrs)
  !Incidental Ingestion
  !CONSTANT ingest = 0.002 !incidental ingestion rate (ug/hr)
  CONSTANT ingest past = 0.01
  CONSTANT ingest current = 0.01
!!Ingestion Exposure
!DISCRETE IngestON
IF (T.LT.tbackground) THEN !if time is less that tbackground,
ingest = ingest past !ingestion exposure equals ingest past
else !if time is less that tbackground,
ingest = ingest current !ingestion exposure equals ingest past
ENDIF
aingestdose = integ(ingest, 0.)
!Drinking water exposure
DISCRETE Drinkint
SCHEDULE DrinkON .AT. tstart
END
DISCRETE DrinkON
IF (T.LT.tbackground) THEN !if time is less than tbackground,
INTERVAL C2 = 4.0 !simulate drinking exposure every four hours
IF (hourofday .LT. 16.0) THEN !if hour of day is less than 16 (8pm)
drinkdose = ((backgrounddw/tlendw)*(dwtotal/drinks)) !simulate drinking water (background
concentration) exposure at set interval(ug/hr)
countdw = countdw + 1 !increase drinking episode count by 1 following exposure
SCHEDULE DrinkOFF .AT. T+tlendw !turn off drinking water exposure after 15 minutes (tlen)
ENDIF
hourofday = hourofday + 4.0 !increase hour of day count by four hours following each exposure
```

```
IF (hourofday .EQ. 24.0) dayofweek = dayofweek + 1.0 !when hour of day count reaches 24, increase day of week
count by 1
IF (hourofday .EQ. 24.0) hourofday = 0.0 !when hour of day count reaches 24, reset hour of day count to zero
IF (dayofweek .EQ. 7.0) dayofweek = 0.0 !when day of week count reaches 7, reset day of week count to zero
ENDIF
END
adrinkdose = integ(drinkdose, 0.)
DISCRETE DrinkON
IF (T.GT.tbackground) THEN !if time is greater than tbackground,
INTERVAL C2 = 4.0 !simulate drinking exposure every four hours
IF (hourofday .LT. 16.0) THEN !if hour of day is less than 16 (8pm)
drinkdose = ((exposeddw/tlendw)*(dwtotal/drinks)) !simulate drinking water (exposed concentration)exposure at
set interval(ug/hr)
countdw = countdw + 1.0 !increase drinking episode count by 1 following exposure
SCHEDULE DrinkOFF .AT. T+tlendw !turn off drinking water exposure after 15 minutes (tlen)
ENDIF
hourofday = hourofday + 4.0 !increase hour of day count by four hours following each exposure
IF (hourofday .EQ. 24.0) dayofweek = dayofweek + 1.0 !when hour of day count reaches 24, increase day of week
count by 1
IF (hourofday .EQ. 24.0) hourofday = 0.0 !when hour of day count reaches 24, reset hour of day count to zero
IF (dayofweek .EQ. 7.0) dayofweek = 0.0 !when day of week count reaches 7, reset day of week count to zero
ENDIF
END
adrinkdose2 = integ(drinkdose, 0.0)
DISCRETE DrinkOFF
Drinkdose = 0.0 !set drink dose to zero during non-exposure periods
END
```

```
!Model Equations
   !Rest of Body (Tis)
   RR = QR*(CA-CVR)*Free !rate of change in rest of body (ug/h)
   AR = integ(RR, 0.0) !amount in rest of body (ug)
   CR = AR/VR !concentration in rest of body (ug/L)
   CVR = CR/PR !concentration in venous blood leaving the rest of the body (ug/L)
   !Kidney
     !Kidney Blood (Kb)
     RKb = QK*(CA-CVK)*Free - CA*GFR*Free - Rdif - RA baso !rate of change in kidney blood (ug/h).
     AKb = integ(RKb, 0.0) !amount in kidney blood (ug)
     CKb = AKb/VKb !concentration in kidney blodd (ug/L)
     CVK = CKb !/PK !concentration in venous blood leaving kidney (ug/L)
     RCl = CA*GFR*Free !rate of clearance via glomerular filtration (ug/h)
     ACl = integ(RCl, 0.0) !amount moved via glomerular filtration (ug)
     Rdif = Kdif*(CKb - CPTC) !rate of diffusion from into the PTC (ug/hr)
     Adif = integ(Rdif, 0.0) !amount diffused into the PTC (ug)
     RA baso = (Vmax baso*CKb)/(Km baso + CKb) !rate of transport through basolateral transporters (ug/h)
     A baso = integ(RA baso, 0.0) !amount transported through basolateral transporters (ug)
      !Proximal Tubule Cells (PTC)
     RPTC = Rdif + RA apical + RA baso - RAefflux !rate of change in PTC(ug/h)
     APTC = integ(RPTC, 0.0) !amount in proximal tubule cells (ug)
     CPTC = APTC/VPTC !concentration in PTC (ug/L)
     RA apical = (Vmax apical*Cfil) / (Km apical + Cfil) !rate of transport through apical transporters
(ug/h)
     A apical = integ(RA apical, 0.0)!amount transported through apical transporters (ug)
     RAefflux = kefflux*APTC   !rate of efflux from proximal tubule cells into circulation (ug/h)
     Aefflux = integ(RAefflux, 0.0) !amount effluxed from proximal tubule cells into circulation (ug)
```

```
!Filtrate (Fil)
  Rfil = CA*GFR*Free - RA apical - Afil*kurine !rate of change in filtrate (ug/h)
 Afil = integ(Rfil, 0.0) !amount in filtrate (ug)
  Cfil = Afil/Vfil !concentration in filtrate (ug/L)
  !Urinary elimination
  Rurine = kurine*Afil !rate of change in urine (ug/h)
  Aurine = integ(Rurine, 0.0) !amount in urine (ug)
  Curine = Rurine/kvoid !concentration in urine (ug/L)
!GI Tract (Absorption site of oral dose)
  !Stomach
 RST= ingest + drinkdose - k0*AST - GE*AST !rate of change in the stomach (ug/h)
 AST = integ(RST, 0.0) !amount in the stomach (ug)
  RabsST = k0*AST !rate of absorption in the stomach (ug/h)
 AabsST = integ(RabsST, 0.0) !amount absorbed in the stomach (ug)
  !Small Intestine
  RSI = GE*AST - kabs*ASI - kunabs*ASI !rate of change in the small intestine (ug/hr)
  ASI = integ(RSI, 0.0) !amount in the small intestine (ug)
  RabsSI = kabs*ASI !rate of absorption in the small intestine (ug/hr)
  AabsSI = integ(RabsSI, 0.0) !amount absorbed in the small intestine (ug)
  total oral uptake = AabsSI + AabsST !total oral uptake in the GI tract (uq)
  !Feces compartment
  Rfeces = kbile*AL + kunabs*ASI !rate of change in the feces compartment (ug/h)
 Afeces = integ(Rfeces, 0.0) !amount in the feces compartment (ug)
!Liver
RL = QL*(CA-CVL)*Free - kbile*AL + kabs*ASI + k0*AST !rate of change in the liver (ug/h)
AL = integ(RL, 0.0) !amount in liver (ug)
CL = AL/VL !concentration in the liver (ug/L)
```

```
CVL = CL/PL !concentration in the venous blood leaving the liver (ug/L)
           CLiver = AL/ML !concentration in liver (ug/g)
           Abile = kbile*AL !amount in the bile (ug)
           !Plasma compartment
           RPlas free = (QR*CVR*Free) + (QK*CVK*Free) + (QL*CVL*Free) - (QC*CA*Free) + RAefflux ! rate of change in the context of the 
the plasma (uq/h)
           Aplas free = integ(RPlas free, 0.0) !amount in the plasma (ug)
           CA_free = APlas_free/VPlas !concentration in plasma (ug)
           CA = CA free/Free !concentration of total PFOA in plasma (ug/L)
            !Mass Balance Check
           Atissue = APlas_free + AR + AKb + AFil + APTC + AL + AST + ASI !sum of mass in all compartments (ug)
           Aloss = Aurine + Afeces !sum of mass lost through urinary and fecal excretion (ug)
           totaldose = Adrinkdose + Aingestdose
                              Atotal = Atissue + Aloss !total mass; should equal total dose
           END ! DERIVATIVE
           CONSTANT TSTOP = 24.0 !hours
           TERMT (T .GE. TSTOP, 'checked on communication interval: REACHED TSTOP')
      END ! DYNAMIC
      TERMINAL.
           ! code that is executed once at the end of a simulation run goes here
     END ! TERMINAL
END ! PROGRAM
```

## **Additional Sensitivity Analysis Information**

Sensitivity coefficients were determined for the serum concentration resulting from a 1% change in the value of each parameter value using the forward difference method. Sensitivity analysis was conducted for simulations with drinking water concentration set to high (3.55  $\mu$ g/L) and low (0.04  $\mu$ g/L) concentrations. Positive sensitivity coefficients indicate a direct association between the model output and the corresponding parameter. Negative sensitivity coefficients indicate an inverse correlation between the model output and the corresponding parameter. Parameters with absolute sensitivity coefficients greater than 0.1 were identified as sensitive.

**Table S1.** Normalized sensitivity coefficients for sensitivity analysis of the PBPK model for PFOA in the adult male human.

PARAMETER	SENSITIVITY	SENSITIVITY
	COEFFICIENT (DW	COEFFICIENT (DW
	$CONC = 3.55 \mu g/L)$	$CONC = 0.04 \mu g/L)$
$\mathbf{BW}$	-0.839478	-0.839478
DRINKS	-0.937227	-0.937227
DWTOTAL	0.946598	0.946598
EXPOSEDDW	0.946599	0.946593
FREE	-0.387236	-0.387236
GEC	0.000064	0.000064
GFRC	-0.388368	-0.388368
HTC	-0.000895	-0.000895
INGEST	0.000000	0.000000
K0C	-0.000091	-0.000091
KABSC	-0.000399	-0.000399
KBILEC	-0.611969	-0.611969
KDIF	0.000005	0.000005
KEFFLUXC	0.000025	0.000025
KM_APICAL	-0.383738	-0.383738
KM_BASO	0.000008	0.000008
KUNABSC	-0.000026	-0.000026
KURINEC	-0.3852010	-0.3851814
KVOID	0.000000	0.000000
PK	0.000000	0.000000
PL	-0.607463	-0.607463
PR	-0.000075	-0.000075
PROTEIN	0.382970	0.382970
QCC	0.001118	0.001118
QKC	-0.000235	-0.000235

QLC	0.000253	0.000253
RAFAPI	0.382978	0.382978
RAFBASO	-0.000008	-0.000008
TLENDW	0.000046	0.000046
VFILC	-0.383738	-0.383738
VKC	-0.004561	-0.004561
VLC	-0.607460	-0.607460
VMAX_APICAL_INVITRO	0.382978	0.382978
VMAX_BASO_INVITRO	-0.000008	-0.000008
VPLASC	0.000015	0.000015
VPTCC	-0.000005	-0.000005